MASSACHVSETTS INSTITVTE OF TECHNOLOGY
Department of Electrical Engineering and Computer Science
6.001—Structure and Interpretation of Computer Programs
Spring 2006

Recitation 4
Orders of Growth

Definitions

Theta (Θ) notation:

$$f(n) = \Theta(g(n)) \rightarrow k_1 \cdot g(n) \leq f(n) \leq k_2 \cdot g(n), \text{for } n > n_0$$

Big-O notation:

$$f(n) = O(g(n)) \rightarrow f(n) \leq k \cdot g(n), \text{for } n > n_0$$

Adversarial approach: For you to show that $f(n) = \Theta(g(n))$, you pick k_1, k_2, and n_0, then I (the adversary) try to pick an n which doesn't satisfy $k_1 \cdot g(n) \leq f(n) \leq k_2 \cdot g(n)$.

Implications

Ignore constants. Ignore lower order terms. For a sum, take the larger term. For a product, multiply the two terms. Orders of growth are concerned with how the effort scales up as the size of the problem increases, rather than an exact measure of the cost.

Typical Orders of Growth

- $\Theta(1)$ - Constant growth. Simple, non-looping, non-decomposable operations have constant growth.
- $\Theta(\log n)$ - Logarithmic growth. At each iteration, the problem size is scaled down by a constant amount: `(call-again (/ n c))`.
- $\Theta(n)$ - Linear growth. At each iteration, the problem size is decremented by a constant amount: `(call-again (- n c))`.
- $\Theta(n \log n)$ - Nifty growth. Nice recursive solution to normally $\Theta(n^2)$ problem.
- $\Theta(n^2)$ - Quadratic growth. Computing correspondence between a set of n things, or doing something of cost n to all n things both result in quadratic growth.
- $\Theta(2^n)$ - Exponential growth. Really bad. Searching all possibilities usually results in exponential growth.

What's n?

Order of growth is *always* in terms of the size of the problem. Without stating what the problem is, and what is considered primitive (what is being counted as a "unit of work" or "unit of space"), the order of growth doesn't have any meaning.

Problems

1. Give order notation for the following:
   (a) $5n^2 + n$
   $\Theta(n^2)$
   (b) $\sqrt{n} + n$
   $\Theta(n)$
   (c) $3^n n^2$
   $\Theta(3^n n^2)$

2. ```
   (define (fact n)
     (if (= n 0)
         1
         (* n (fact (- n 1)))))
   ```
   Running time? $\Theta(n)$ Space? $\Theta(n)$

3. ```
   (define (find-e n)
     (if (= n 0)
         1.
         (+ (/ (fact n)) (find-e (- n 1)))))
   ```
   Running time? $\Theta(n^2)$ Space? $\Theta(n)$

4. Assume you have a procedure `(divisible? n x)` which returns `#t` if `n` is divisible by `x`. It runs in $O(n)$ time and $O(1)$ space. Write a procedure `prime?` which takes a number and returns `#t` if it's prime and `#f` otherwise. You'll want to use a helper procedure.

   ```
   (define (prime? p)
      (define (helper n)
         (if (> n (sqrt p))
             #t
             (if (divisible? p n)
                #f
                (helper (+ n 1)))))
      (helper 2))

   ; iterative process
   ; assuming sqrt takes O(1) time, sqrt(n) * n
   ```
   Running time? $\Theta(n\sqrt{n})$ Space? $\Theta(1)$

5. Write an iterative version of `find-e`. What is its running time and space?

   ```
   (define (find-e-iter n)
     (define (helper n s)
         (if (= n 0) s
           (helper (- n 1) (+ s (/ (fact n))))))
     (helper n 1.0))
   ```
   Running time? $\Theta(n^2)$ Space? $\Theta(n)$

Micro Quiz

Name:

1. Write a procedure that computes the number of decimal digits in it's input. Do not use logs. You may use `quotient` (integer division) if you wish.
`(num-digits 102)` → `3`

```
(define (num-digits n)
  (if (= n 0)
      0
      (+ 1 (num-digits (quotient n 10)))))
```

Running time? $\Theta(\log n)$ Space? $\Theta(\log n)$

2. Write a procedure that will multiply two positive integers together, but the only arithmetic operation allowed is addition (ie multiplication through repeated addition). In addition, your procedure should be iterative, not recursive.
`(slow-mul 3 4)` → `12`

```
(define (mul-helper a b total)
  (if (= a 0)
      total
      (mul-helper (- a 1) b (+ total b)))) ; or (+ a -1) if picky
 (define (slow-mul a b)
  (mul-helper a b 0))
```

Running time? $\Theta(n)$ Space? $\Theta(1)$